

MSX BIOS

Copyrighted © 1985 by ASCII Corporation of Japan

All rights Reserved

Edited: January 1985
by Steven M. Ting
Graphic design: Mervin Fong.

The information in this document is subject to change without notice. ASCII Corp. makes no warranty with regard to the manual, including but not limited to, implied warranties of merchantability and fitness for a particular purpose. The parties above assume no responsibility for any errors which may appear in this document.

This document is not intended as "Consumer goods" under applicable federal or state law(s).

No part of this document may be copied or reproduced in any form or by any means without the prior written consent of ASCII Corporation and Qest Publishing Inc.

MSX is a registered trademark of Microsoft Corporation, Bellevue, WA.

Z80 is a registered trademark of Zilog, Inc.

Printed in United States

Scanned and converted to PDF by HansO, 2003

TABLE OF CONTENTS

BIOS LISTING	1 - 256
MSX BIOS CROSS REFERENCE.....	257 - 280
SYMBOL TABLE.....	281 - 285
APPENDIX A	
MSX USA & UK OVERLAY PATCHES.....	287 - 316
BIOS CALLS.....	317 - 324
APPENDIX B	
CHARACTER SET & KEYBOARD LAYOUT.....	325 - 338
HOOKS & RAM ROUTINES.....	339 - 356

```

1          .list
2          ;
3          ;
4          ; (C) Copyright by ASCII Corp., 1983
5          ; Proprietary information. All rights reserved.
6          ;
7          ; File: BIOHDR.MAC
8          ; USE: Restart calls and ROM entries table
9          ; Written by Jey Suzuki, Rick Yamashita
10         ; ASCII Corporation, Japan
11         ;
12         ; Edit: January, 1985
13         ; Reason: Zilog Z80 Mnemonic version and cleanup
14         ; Edited by: Steven M. Ting
15         ;
16         ;
17         ; Labels referenced in this listing, are the absolute locations
18         ; within the MSX ROM. However, "ONLY" this BIOS entry point table,
19         ; and RAM variables are guaranteed to be permanent.
20         ;
21         ; All other locations in the ROM, will be changed without notice.
22         ;
23         SUBTTL -BIOS header- BIOS calls (Basic Interpreter, Slot I/O)
  
```

```

24
25         ;
26         ; The following RST's (RST 0 thru RST 5) are reserved for BASIC
27         ; interpreter, RST 6 for inter-slot calls, and RST 7 for
28         ; hardware interrupt
29         ;
30         0000 F3 BEGIN: DI ;Fail safe
31         0001 C3 02D7 JP CHKRAM ;Finds all connected RAM
32         ; and cartridges
33         ;
34         ;
35         ; ** Special information for the VDP. **
36         ; Any program that accesses the VDP hardware directly
37         ; should read the I/O port address found here, to be certain
38         ; the software is compatible with future versions of the VDP.
39         ;
40         0004 1BBF DW CGTABL ;Address of character generator table,
41         ; to allow use of other character ROM.
42         ;
43         0006 98 DB 98H ;Current port address for VDP Data read
44         0007 98 DB 98H ; " " " " " " write
45         ;
46         0008 C3 2683 JP SYNCHR ;Check byte following the RST 8, see
47         ; if equal to the byte pointed by HL
48         000B 00 DB 0
49         000C C3 01B6 JP RDSLTL ;Read a byte from another slot
50         000F 00 DB 0
51         0010 C3 2686 JP CHRCTR ;Fetch next char from BASIC text
52         0013 00 DB 0
53         0014 C3 01D1 JP WRSLTL ;Write a byte to another slot
54         0017 00 DB 0
  
```

```

( MSX ROM BASIC BIOS ) Macro-80      3.44  01-Jan-85      PAGE  2-1      3
-BIOS header- BIOS calls (Basic Interpreter, Slot I/O)

55  0018  C3 1B45      JP  OUTDO      ;Output a char to the Console or printer
56  001B  00          DB  0            ;
57  001C  C3 0217      JP  CALSLT     ;Perform Inter-slot call
58  001F  00          DB  0            ;
59  0020  C3 146A      JP  DCOMPR     ;Compares [HL] to [DE]
60  0023  00          DB  0            ;
61  0024  C3 025E      JP  ENASLT     ;Permanently enables a slot
62  0027  00          DB  0            ;
63  0028  C3 2689      JP  GETYPR     ;Returns the [FAC] type
64  002B  00          DB  0            ;ID Byte (1)
65          ;Format:
66          ; B7 B6 B5 B4 B3 B2 B1 B0
67          ; + + + + + + + +
68          ; + + + + - - - - - Type of character
69          ; + + + + generator.
70          ; + + + + 0:Japanese
71          ; + + + + 1:International
72          ; + + + + 2:Korea
73          ; + + - - - - - Date format
74          ; + + 0: Y-M-D 1: M-D-Y
75          ; + + 2: D-M-Y
76          ; - - - - - Interrupt frequency
77          ; 0: 60 Hz 1: 50 Hz
78  002C  00          DB  0            ;ID Byte (2)
79          ;Format:
80          ; B7 B6 B5 B4 B3 B2 B1 B0
81          ; + + + + + + + +
82          ; + + + + - - - - - Type of Keyboard
83          ; + + + + 0:Japanese 2:French
84          ; + + + + 1:Int 3:UK
85          ; + + + + 4:DIN

```

```

( MSX ROM BASIC BIOS ) Macro-80      3.44  01-Jan-85      PAGE  2-2      4
-BIOS header- BIOS calls (Basic Interpreter, Slot I/O)

86          ; - - - - - Version of BASIC
87          ; 0: Japanese
88          ; 1: International
89  002D  00 00 00      DB  0,0,0
90  0030  C3 0205      JP  CALLF     ;Performs Far-call (i.e., Inter-slot)
91  0033  00 00 00 00  DB  0,0,0,0,0
92  0037  00
93          ;
94          ;
95          ; Following are used for I/O initialization
96          ;
97  0038  C3 0C3C      JP  KEYINT     ;Handlers for hardware interrupt
98  003B  C3 049D      JP  INITIO     ;Do device initialization
99  003E  C3 139D      JP  INIFNK     ;Reset all function key's text
100         ;
101         SUBTTL -BIOS header- BIOS calls (Video display processor)

```

```
102
103 ;
104 ; The following entry points provides control of the
105 ; VDP's registers, screen mode settings, and memory block
106 ; move between DRAM and VRAM.
107 ;
108 0041 C3 0577 JP DISSCR ;Disables screen display
109 0044 C3 0570 JP ENASCR ;Enables screen display
110 0047 C3 057F JP WRTVDP ;Write a byte to any VDP register
111 004A C3 07D7 JP RDVRM ;Read VRAM addressed using [HL]
112 004D C3 07CD JP WRTVRM ;Write VRAM addressed using [HL]
113 0050 C3 07EC JP SETRD ;Sets up VDP for read
114 0053 C3 07DF JP SETWRT ;Sets up VDP for write
115 0056 C3 0815 JP FILVRM ;Fills VRAM with specified data
116 0059 C3 070F JP LDIRMV ;Moves block of data from VRAM to memory
117 005C C3 0744 JP LDIRVM ; " " " " " memory to VRAM
118 005F C3 084F JP CHGMOD ;Change screen mode of VDP to [SCRMOD]
119 0062 C3 07F7 JP CHGCLR ;change Foreground, background,
120 ;border, color
121 0065 00 DB 0
122 ;
123 ;
124 0066 C3 1398 JP NMI ;Handler for non-maskable interrupt
125 ;
126 0069 C3 06A8 JP CLRSRP ;Init sprite data
127 006C C3 050E JP INITXT ;Init VDP for 40 X 24 text mode (SCREEN 0)
128 006F C3 0538 JP INIT32 ; " " " 32 X 24 text mode (SCREEN 1)
129 0072 C3 05D2 JP INIGRP ; " " " High resolution mode (SCREEN 2)
130 0075 C3 061F JP INIMLT ; " " " Multi color mode (SCREEN 3)
131 0078 C3 0594 JP SETTXT ;Sets VDP to display 40 X 24 text mode
132 007B C3 05B4 JP SETT32 ; " " " " 32 X 24 text mode
```

```
133 007E C3 0602 JP SETGRP ; " " " " High-res mode
134 0081 C3 0659 JP SETMLT ; " " " " Multi color mode
135 0084 C3 06E4 JP CALPAT ;Get address of sprite pattern table
136 0087 C3 06F9 JP CALATR ; " " " " attribute table
137 008A C3 0704 JP GSPSIZ ;Returns current sprite size
138 008D C3 1510 JP GRPPRT ;Print a character on the graphic screen
139 ;
140 SUBTTL -BIOS header- BIOS calls (Programmable Sound Generator control)
```

```
141
142 ;
143 ; Following entry points are used for PSG initialization,
144 ; read and write PSG registers, and PLAY statement execution.
145 ;
146 0090 C3 04BD JP GICINI ;Init PSG, and static data for PLAY
147 0093 C3 1102 JP WRTPSG ;Write data to PSG
148 0096 C3 110E JP RDPSG ;Read data from PSG
149 0099 C3 11C4 JP STRTMS ;Checks and start background task for PLAY
150 ;
151 SUBTTL -BIOS header- BIOS calls (Keyboard, CRT, and Printer)
```

```
152
153 ;
154 ; General INPUT and PRINT utilities.
155 ;
156 009C C3 0D6A JP CHSNS ;Checks status of keyboard status
157 009F C3 10CB JP CHGET ;Return char typed, with wait
158 00A2 C3 08BC JP CHPUT ;Output character to console
159 00A5 C3 085D JP LPTOUT ; " " to printer, if possible
160 00A8 C3 0884 JP LPTSTT ;Checks status of line printer
161 00AB C3 089D JP CNVCHR ;Checks for graphic header byte
162 ;and convert code
163 00AE C3 23BF JP PINLIN ;Read line from keyboard to buffer
164 00B1 C3 23D5 JP INLIN ;Same as above, except in case of
165 ;AUTFLG is set
166 00B4 C3 23CC JP QINLIN ;Print a "?", then jump to INLIN
167 00B7 C3 046F JP BREAKX ;[Control-STOP] pressed??
168 00BA C3 03FB JP ISCNTC ;[Shift-STOP] pressed??
169 00BD C3 10F9 JP KKCNTC ;Same as ISCNTC, but used by BASIC
170 00C0 C3 1113 JP BEEP ;Buzz
171 00C3 C3 0848 JP CLS ;Clear screen
172 00C6 C3 088E JP POSIT ;Place cursor at Column [H], Row [L]
173 00C9 C3 0B26 JP FNKSB ;Display Function key, if necessary
174 00CC C3 0B15 JP ERAFNK ;Stop displaying the Function keys
175 00CF C3 0B2B JP DSPFNK ;Enable Function key display
176 00D2 C3 083B JP TOTEXT ;Force screen to text mode
177 ;
178 SUBTTL -BIOS header- BIOS calls (Game and Cassette I/O, Queue handler)
```

```
179
180 ;
181 ; Following are used to read the value from Joysticks,
182 ; Graphic pad (tablet), and Paddles.
183 ;
184 00D5 C3 11EE JP GTSTCK ;Return status of joystick
185 00D8 C3 1253 JP GTTRIG ;Read joystick trigger button
186 00DB C3 12AC JP GTPAD ;Returns status of graphic pad
187 00DE C3 1273 JP GTPDL ;Read paddle
188 ;
189 ;
190 ; Following are used to access the cassette tape,
191 ; data read/write, and motor on/off
192 ;
193 00E1 C3 1A63 JP TAPION ;Turn on motor and read tape header
194 00E4 C3 1ABC JP TAPIN ;Read tape data
195 00E7 C3 19E9 JP TAPIOP ;Stops reading from tape
196 00EA C3 19F1 JP TAPOON ;Turn on motor and write tape header
197 00ED C3 1A19 JP TAPOUT ;Write data to tape
198 00F0 C3 19DD JP TAPOFF ;Stops writing to tape
199 00F3 C3 1384 JP STMOTR ;Start, stop cassette motor, or
200 ;flip motor(on to off, off to on)
201 ;
202 ;
203 ; BASIC queues
204 ;
205 00F6 C3 14EB JP LFTQ ;Bytes left in queue
206 00F9 C3 1492 JP PUTQ ;Send a byte to queue
207 ;
208 SUBTTL -BIOS header- BIOS calls (Generalized graphics)
```

```

209
210
211 ;
212 ; For BASIC interpreter's GENGPR and ADVGRP modules use
212 00FC C3 16C5 JP RIGHTC ;Moves one pixel right
213 00FF C3 16EE JP LEFTC ; " " " left
214 0102 C3 175D JP UPC ; " " " up
215 0105 C3 173C JP TUPC ; " " " "
216 0108 C3 172A JP DOWNC ; " " " down
217 010B C3 170A JP TDOWNC ; " " " "
218 010E C3 1599 JP SCALXY ;Scales X Y coordinates
219 0111 C3 15DF JP MAPXYC ;Maps coordinates to physical address
220 0114 C3 1639 JP FETCHC ;Get current physical address and
221 ;mask pattern
222 0117 C3 1640 JP STOREC ;Put current physical address and
223 ;mask pattern
224 011A C3 1676 JP SETATR ;Sets the color attribute byte
225 011D C3 1647 JP READC ;Reads attribute of current pixel
226 0120 C3 167E JP SETC ;Sets current pixel to specified attribute
227 0123 C3 1809 JP NSETCX ;Sets pixel horizontally
228 0126 C3 18C7 JP GTASPC ;Returns aspect ratio
229 0129 C3 18CF JP PNTINI ;Do paint initialization
230 012C C3 18E4 JP SCANR ;Scan pixels to the right
231 012F C3 197A JP SCANL ; " " " " left
232 ;
233 SUBTTL -BIOS header- BIOS calls (Misc. Entries)
  
```

```

234
235 ;
236 ;
237 0132 C3 0F3D JP CHGCAP ;Turn [CAPSLOCK] light, on/off
238 0135 C3 0F7A JP CHGSND ;Change status of 1 bit sound port
239 0138 C3 144C JP RSLREG ;Return output of primary slot register
240 013B C3 144F JP WSLREG ;Write to primary slot register
241 013E C3 1449 JP RDVDP ;Read VDP status register
242 0141 C3 1452 JP SNSMAT ;Read a specified row in the
243 ;keyboard matrix
244 0144 C3 148A JP PHYDIO ;Performs operation for mass storage
245 ;devices (such as disks)
246 0147 C3 148E JP FORMAT ;Initialize mass storage device
247 014A C3 145F JP ISFLIO ;Are we doing device I/O
248 014D C3 1B63 JP OUTDLP ;Output to line printer
249 0150 C3 1470 JP GETVCP ;Used by Music background tasking
250 0153 C3 1474 JP GETVC2 ; " " " " "
251 0156 C3 0468 JP KILBUF ;Clear the keyboard buffer
252 0159 C3 01FF JP CALBAS ;Performs far-call into BASIC
253 015C DS 005AH ;RESERVED FOR EXPANSION
254 ;
255 SUBTTL - SLOT - Slot handler stuff
  
```

```

256
257 00A8 PPI.AR EQU 0A8h ;A8H read from PPI Port A
258 00A8 PPI.AW EQU 0A8h ;A8H Write to PPI Port A
259 ;
260 ; Every cartridge located at 0000-3FFH must contain codes in
261 ; this module which are entered via following addresses.
262 ;
263 ; 000CH RDSLT
264 ; 0014H WRSLT
265 ; 001CH CALSLT
266 ; 0024H ENASLT
267 ;
268 ;
269 ; ----- RDSLT -----
270 ;
271 ; Selects the appropriate slot according to the value given
272 ; through registers, and read the content of memory from the
273 ; slot.
274 ;
275 ; Input parameters:
276 ; A - FxxxSSPP
277 ; | |||
278 ; | +++- primary slot # (0-3)
279 ; | +---- secondary slot # (0-3)
280 ; +----- 1 if secondary slot # specified
281 ;
282 ; HL - address of target memory
283 ; Returned value
284 ; A - content of memory
285 ;
286 ; Note: Interrupts are disabled automatically but never enabled
  
```

```

287 ; by this routine.
288 ;
289 ; RDSLT:
290 01B6 CD 027E CALL SELPRM ;Calculate bit pattern and mask code
291 01B9 FA 01C6 JP M,RDESLT ;Expanded slot specified
292 01BC DB A8 IN A,(PPI.AR)
293 01BE 57 LD D,A ;Save current setting
294 01BF A1 AND C ;Cancel current setting for target address
295 01C0 B0 OR B ;Add new setting
296 01C1 CD F380 CALL RAMLOW ;Call read primitive routine (in system area)
297 01C4 7B LD A,E ;Return value via [Acc]
298 01C5 C9 RET
299 01C6 RDESLT:
300 01C6 E5 PUSH HL ;Save target address
301 01C7 CD 02A3 CALL SELEXP ;Select secondary slot
302 01CA E3 EX (SP),HL ;Restore target address and save [HL]
303 01CB C5 PUSH BC
304 01CC CD 01B6 CALL RDSLT
305 01CF 18 1B JR WRESED ;Restore old slot select register
306 SUBTTL -SLOT- Slot handler (Write slot)
  
```

```

307 ;
308 ; ----- WRSLOT -----
309 ;
310 ; Selects the appropriate slot according to the value given
311 ; through registers, and write to the memory in the specified
312 ; slot.
313 ;
314 ;
315 ; Input parameters:
316 ; A - FxxxSSPP
317 ; | |||
318 ; | ||+--- primary slot # (0-3)
319 ; | +----- secondary slot # (0-3)
320 ; +----- 1 if secondary slot # specified
321 ;
322 ; HL - address of target memory
323 ;
324 ; E - value to be written
325 ;
326 ; Note: Interrupts are disabled automatically but never enabled
327 ; by this routine.
328 ;
329 01D1 WRSLOT:
330 01D1 D5 PUSH DE ;Save data to be written
331 01D2 CD 027E CALL SELPRM ;Calculate bit pattern and mask code
332 01D5 FA 01E1 JP M,WRESLT ;Expanded slot specified
333 01D8 D1 POP DE ;Restore data to be written
334 01D9 DB A8 IN A,(PPI.AR)
335 01DB 57 LD D,A ;Save current setting
336 01DC A1 AND C ;Cancel current setting for target address
337 01DD B0 OR B ;Add new setting
  
```

```

338 01DE C3 F385 JP WRPRIM ;Call write primitive routine (in system area)
339 01E1 WRESLT:
340 01E1 E3 EX (SP),HL ;Save target address, get data to be written
341 01E2 E5 PUSH HL ;Save data to be written
342 01E3 CD 02A3 CALL SELEXP ;Select secondary slot
343 01E6 D1 POP DE ;Restore data to be written
344 01E7 E3 EX (SP),HL ;Restore target address and save [HL]
345 01E8 C5 PUSH BC
346 01E9 CD 01D1 CALL WRSLOT
347 01EC WRESED:
348 01EC C1 POP BC
349 01ED E3 EX (SP),HL ;Save target address and get old [HL]
350 01EE F5 PUSH AF ;Save value returned by RDSLT
351 01EF 78 LD A,B ;Get current setting
352 01F0 E6 3F AND 00111111B ;Cancel current setting for 0C000H..0FFFFH
353 01F2 B1 OR C
354 01F3 D3 A8 OUT (PPI.AW),A ;Enable 0C000H..0FFFFH of target bank
355 01F5 7D LD A,L ;Restore old setting of slot register
356 01F6 32 FFFF LD (0FFFFH),A
357 01F9 78 LD A,B ;Finally restore old primary slot register
358 01FA D3 A8 OUT (PPI.AW),A
359 01FC F1 POP AF ;Restore value returned by RDSLT
360 01FD E1 POP HL ;Restore target address
361 01FE C9 RET
  
```



```
362
363      01FF          CALBAS: LD      IY,(EXPTBL-1)
364      01FF  FD 2A FCC0      JR      CALSLT
365      0203  18 12
366      0205          CALLF:
367      0205  E3            EX      (SP),HL      ;Get return address, save [HL]
368      0206  F5            PUSH   AF          ;Save working registers
369      0207  D5            PUSH   DE
370      0208  7E            LD      A,(HL)      ;Get destination slot
371      0209  F5            PUSH   AF
372      020A  FD E1          POP     IY          ;Move it to IYH
373      020C  23            INC     HL
374      020D  5E            LD      E,(HL)      ;Get destination address
375      020E  23            INC     HL
376      020F  56            LD      D,(HL)
377      0210  23            INC     HL          ;Prepare true return address
378      0211  D5            PUSH   DE
379      0212  DD E1          POP     IX          ;Move it to IX
380      0214  D1            POP     DE          ;Restore working registers
381      0215  F1            POP     AF
382      0216  E3            EX      (SP),HL      ;Resture [HL], save true return address
383      SUBTTL -SLOT-
```

```
384
385      ;
386      ; ----- CALSLT -----
387      ;
388      ; Performs inter-slot call to specified address.
389      ;
390      ; Input parameters:
391      ; IY - FxxxSSPP
392      ;      | ||||
393      ;      | ||+-- primary slot # (0-3)
394      ;      | ++--- secondary slot # (0-3)
395      ;      +----- 1 if secondary slot # specified
396      ;
397      ; IX - address to call
398      ;
399      ; Note: Interrupts are disabled automatically but never enabled
400      ;       by this routine.
401      ;       You can never pass arguments via alternate registers
402      ;       of Z80.
403      ;
404      0217          CALSLT:
405      0217  D9            EXX          ;Save environments
406      0218  08            EX      AF,AF'
407      0219  FD E5          PUSH   IY
408      021B  F1            POP     AF          ;Get target slot information
409      021C  DD E5          PUSH   IX
410      021E  E1            POP     HL          ;Get target address
411      021F  CD 027E        CALL   SELPRM
412      0222  FA 022E        JP     M,CALLESL      ;Call expanded slot
413      0225  DB A8          IN     A,(PPI.AR)
414      0227  F5            PUSH   AF          ;Save current value of primary slot register
```

```

-SLOT-
415 0228 A1 AND C ;Cancel current setting for target address
416 0229 B0 OR B ;Add new setting
417 022A D9 EXX ;Restore environments except PSW
418 022B C3 F38C JP CLPRIM ;Jump to primitive routine (in system area)
419 022E CALESL:
420 022E CD 02A3 CALL SELEXP ;Select secondary slot register
421 0231 P5 PUSH AF ;Move primary slot # in [IYH]
422 0232 FD E1 POP IY
423 0234 E5 PUSH HL ;Save [B,C,L] which contain information
424 0235 C5 PUSH BC ;for restoring slot environments
425 0236 4F LD C,A ;Move primary slot # to [BC]
426 0237 06 00 LD B,0
427 0239 7D LD A,L ;Re-calculate what is currently output
428 023A A4 AND H ;to expansion slot register
429 023B B2 OR D
430 023C 21 FCC5 LD HL,SLTTBL ;Calculate address into SLTTBL
431 023F 09 ADD HL,BC
432 0240 77 LD (HL),A ;Set current value output to expansion
433 ;slot register
434 0241 E5 PUSH HL ;Remember this address
435 0242 08 EX AF,AF' ;Restore possible arguments passed
436 0243 D9 EXX ;via registers
437 0244 CD 0217 CALL CALSLT ;Call by primary slot #
438 0247 D9 EXX ;Save possible values returned via
439 0248 08 EX AF,AF' ;registers
440 0249 E1 POP HL ;Restore address into SLTTBL
441 024A C1 POP BC ;Restore information about old slots
442 024B D1 POP DE
443 024C 78 LD A,B ;Get current setting
444 024D E6 3F AND 0011111B ;Cancel current setting for 0C000H..0FFFFH
445 024F B1 OR C

```

```

-SLOT-
446 0250 F3 DI
447 0251 D3 A8 OUT (PPI.AW),A ;Enable 0C000H..0FFFFH of target bank
448 0253 7B LD A,E ;Restore old setting of slot register
449 0254 32 FFFF LD (0FFFFH),A
450 0257 78 LD A,B ;Finally restore old primary slot register
451 0258 D3 A8 OUT (PPI.AW),A
452 025A 73 LD (HL),E ;And change SLTTBL also
453 025B 08 EX AF,AF' ;Restore possible returned values
454 025C D9 EXX
455 025D C9 RET

```

```

-SLOT-
456
457 ;
458 ; ----- ENASLT -----
459 ;
460 ; Selects the appropriate slot according to the value given
461 ; through registers, and permanently enables the slot.
462 ;
463 ; Input parameters:
464 ;
465 ; A - FxxxSSPP
466 ; | |||
467 ; | ||+--- primary slot # (0-3)
468 ; | +----- secondary slot # (0-3)
469 ; +----- 1 if secondary slot # specified
470 ;
471 ; HL - address of target memory
472 ;
473 ; Note: Interrupts are disabled automatically but never enabled
474 ; by this routine.
475 ;
476 025E ENASLT:
477 025E CD 027E CALL SELPRM ;Calculate bit pattern and mask code
478 0261 FA 026B JP M,ENESLT ;Expanded slot specified
479 0264 DB A8 IN A,(PPI.AR)
480 0266 A1 AND C ;Cancel current setting for target address
481 0267 B0 OR B ;Add new setting
482 0268 D3 A8 OUT (PPI.AW),A
483 026A C9 RET
484 026B ENESLT:
485 026B E5 PUSH HL ;Save target address
486 026C CD 02A3 CALL SELEXP ;Select secondary slot

```

```

-SLOT-
487 026F 4F LD C,A ;Move primary slot # to [BC]
488 0270 06 00 LD B,0
489 0272 7D LD A,L ;Re-calculate what is currently output
490 0273 A4 AND H ;to expansion slot register
491 0274 B2 OR D
492 0275 21 FCC5 LD HL,SLTTBL ;Calculate address into SLTTBL
493 0278 09 ADD HL,BC
494 0279 77 LD (HL),A ;Set current value output to expansion
495 ;slot register
496 027A E1 POP HL ;Restore target address
497 027B 79 LD A,C ;Restore primary slot # to [Acc]
498 027C 18 E0 JR ENASLT ;Enable by primary slot register
    
```

```

-SLOT-
499
500 027E SELPRM: DI
501 027E F3 PUSH AF ;Save slot address
502 027F F5 LD A,H ;Extract upper 2 bits
503 0280 7C RLCA
504 0281 07 RLCA
505 0282 07 RLCA
506 0283 E6 03 AND 0000011B
507 0285 5F LD E,A
508 0286 3E C0 LD A,0C0H ;Format mask pat. correspond to address
509 0288 SELPRM1: RLCA
510 0288 07 RLCA
511 0289 07 DEC E
512 028A 1D JP P,SELPRM1
513 028B F2 0288 LD E,A ;Save mask pattern
514 028E 5F ; 0000011 0000-3FFF
515 ; 00001100 4000-7FFF
516 ; 00110000 8000-BFFF
517 ; 11000000 C000-FFFF
518
519 028F 2F CPL
520 0290 4F LD C,A ;Save mask pattern
521 ; 11111100 0000-3FFF
522 ; 11110011 4000-7FFF
523 ; 11001111 8000-BFFF
524 ; 00111111 C000-FFFF
525 0291 F1 POP AF ;Restore slot address
526 0292 F5 PUSH AF
527 0293 E6 03 AND 0000011B ;Extract primary slot #
528 0295 3C INC A
529 0296 47 LD B,A
    
```

```

-SLOT-
530 0297 3E AB LD A,10101011B ;Convert slot # to proper bit pattern
531 0299 SELPRM2:
532 0299 C6 55 ADD A,01010101B
533 029B 10 FC DJNZ SELPRM2
534 029D 57 LD D,A ;Save bit pattern for primary slot #
535 ; 00000000 slot #0
536 ; 01010101 slot #1
537 ; 10101010 slot #2
538 ; 11111111 slot #3
539 029E A3 AND E ;Extract significant bits
540 029F 47 LD B,A ;Set it to [B]
541 02A0 F1 POP AF ;Expanded slot specified?
542 02A1 A7 AND A ;Set sign flag if so
543 02A2 C9 RET
544 02A3 SELEXP:
545 02A3 F5 PUSH AF ;Save target slot
546 02A4 7A LD A,D ;Get bit pattern for primary slot
547 02A5 E6 C0 AND 11000000B ;Extract slot # for 0C000H..0FFFFH
548 02A7 4F LD C,A ;Save it
549 02A8 F1 POP AF ;Restore target slot
550 02A9 F5 PUSH AF ;Save target slot
551 02AA 57 LD D,A ;Load [D] with specified slot address
552 02AB DB A8 IN A,(PPI.AR)
553 02AD 47 LD B,A ;Save current setting
554 02AE E6 3F AND 00111111B ;Cancel current setting for 0C000H..0FFFFH
555 02B0 B1 OR C
556 02B1 D3 A8 OUT (PPI.AW),A ;Enable 0C000H..0FFFFH or target bank
557 02B3 7A LD A,D ;Load slot information
558 02B4 0F RRCA
559 02B5 0F RRCA
560 02B6 E6 03 AND 0000011B ;Extract secondary slot #
    
```

```

561 02B8 57          LD      D,A
562 02B9 3E AB      LD      A,10101011B ;Convert secondary slot # to proper
563 02BB          SLEXP1: ADD     A,01010101B ;bit pattern
564 02BB C6 55      DEC     D
565 02BD 15          JP      P,SLEXP1 ; 00000000 slot #0
566 02BE F2 02BB    ; 01010101 slot #1
567          ; 10101010 slot #2
568          ; 11111111 slot #3
569          ;Make bit pattern to be added
570 02C1 A3          LD      D,A
571 02C2 57          LD      A,E
572 02C3 7B          CPL     A,E ;Make bit pattern to strip off old value
573 02C4 2F          LD      H,A
574 02C5 67          LD      A,(0FFFFH) ;Save this
575 02C6 3A FFFF    ;Read expanded slot register
576 02C9 2F          LD      L,A
577 02CA 6F          AND     H ;Save current setting
578 02CB A4          OR      D ;Strip off old bits
579 02CC B2          LD      (0FFFFH),A ;And set new bits
580 02CD 32 FFFF    ;Set secondary slot register
581 02D0 78          LD      A,B
582 02D1 D3 A8      OUT     (PPI.AW),A ;Restore original primary port
583 02D3 F1          POP     AF ;Restore target slot
584 02D4 E6 03      AND     00000011B ;Fake read from primary slot
585 02D6 C9          RET
586          SUBTTL - MSXIO - I/O Module
  
```

```

587          ;
588          ;
589          ;
590          ; Port definition ;
591          ;
592          ;
593          ;
594          ; VDP address definition ;
595          ;
596 0098 VDP.DRW EQU 10011000B ;98H Read/write data VDP
597 0099 VDP.CW EQU 10011001B ;99H write command to VDP
598 0099 VDP.SR EQU 10011001B ;99H read status from VDP
599          ;
600 0007 V.COLR EQU 7 ;In text mode, foreground and background color
601          ; Otherwise background color
602          ;
603          ; PSG address definition ;
604          ;
605 00A0 PSG.LW EQU 10100000B ;A0H latch address for PSG
606 00A1 PSG.DW EQU 10100001B ;A1H write data to PSG
607 00A2 PSG.DR EQU 10100010B ;A2H read data from PSG
608          ;
609 000E PSG.PA EQU 14 ;Port A of PSG
610 000F PSG.PB EQU 15 ;Port B of PSG
611          ;
612          ; PPI address definition ;
613          ;
614 00A8 PPI.AR EQU 10101000B ;A8H read from PPI Port A
615 00A9 PPI.BR EQU 10101001B ;A9H read from PPI Port B
616 00AA PPI.CR EQU 10101010B ;AAH read from PPI Port C
617 00A8 PPI.AW EQU 10101000B ;A8H Write to PPI Port A
  
```

```
618 00AA PPI.CW EQU 10101010B ;AAH write to PPI Port C
619 00AB PPI.CM EQU 10101011B ;ABH write to PPI command register
620 ;
621 ; Printer port definition
622 ;
623 0091 LPT.DW EQU 10010001B ;Data port
624 0090 LPT.SB EQU 10010000B ;Strobe output
625 0090 LPT.ST EQU 10010000B ;Printer status
626 ;
627 ; Text mode (40*24) SCREEN 0
628 ;
629 ; TXTNAM,TXTCGP
630 ;
631 ; Text mode (graphics 1) SCREEN 1
632 ;
633 ; T32NAM,T32COL,T32CGP,T32ATR,T32PAT
634 ;
635 ; Hires mode SCREEN 2
636 ;
637 ; GRPNAM,GRPCOL,GRPCGP,GRPATR,GRPPAT
638 ;
639 ; Multi-color mode SCREEN 3
640 ;
641 ; MLTNAM,MLTCGP,MLTATR,MLTPAT
642 ;
643 ; Screen size
644 ;
645 ; LINLEN,CRTCNT,LINL32,LINL40
646 ;
647 ; External constants
648 ;
```

```
649 ; CGTABL Character generator table
650 ;
651 ; External variables
652 ;
653 ; FORCLR Foreground color
654 ; BAKCLR Background color
655 ; BDRCLR Border color for PAINT
656 ; SCRMOD Current screen mode
657 ; 0 - 40*24 text
658 ; 1 - 32*24 text
659 ; 2 - hiresolution graphics
660 ; 3 - Multicolor graphics
661 ;
662 ; OLDSCR
663 ; NAMBAS Base of current name table
664 ; CGPBAS Base of current cgen table
665 ; PATBAS Base of current sprite pattern table
666 ; ATRBAS Base of current sprite attribute table
667 ; JIFFY Jiffy count
668 ; CLIKSW Click switch
669 ; RGOSAV VDP register #0 save area
670 ; RGLSAV VDP register #1 save area
671 ; STATFL VDP status register
672 ; PATWRK Work area for pattern converter
673 ;
674 ; External routines
675 ;
676 ; GETQ
677 ; PUTQ
678 ; INITQ
679 SUBTTL - MSXIO - Find available RAM
```

```
680
681 02D7          CHKRAM:
682              ;
683              ; ----- CHKRAM -----
684              ;
685              ; Look into every slot from 0FFFFH to C000H, and set system work
686              ; area. Note that we cannot use RAM as work area nor perform
687              ; subroutine calls 'cause we do not yet know where the available
688              ; RAM exits. Everything has to be done inside ROM and CPU's
689              ; register until the RAM is found.
690              ;
691 02D7  3E 82          LD      A,82H          ;Port A - output (mode 0)
692 02D9  D3 AB         'OUT   (PPI.CM),A      ;Port B - input (mode 0)
693 02DB  AF            XOR      A          ;Port C - output (mode 0)
694 02DC  D3 A8         OUT     (PPI.AW),A      ;Select slot 0 for all addresses
695 02DE  3E 50         LD      A,'P'          ;Disable all cassette related outputs
696 02E0  D3 AA         OUT     (PPI.CW),A      ;Motor off
697
698              ; Start searching
699              ;
700              ; Register usage:
701              ; B - non 0 if we're now checking secondary slot
702              ; SPH - slot # of the biggest RAM block
703              ; SPL - secondary slot # of the biggest RAM block (if any)
704              ; DE - lowest address of the biggest RAM block ever found
705              ; C - 'slot-expanded' flag
706              ;
707              ; 0000xxxx
708              ; ||||
709              ; |||+-- slot #3 expanded
710              ; ||+-- slot #2 expanded
```

```
711              ; |+--- slot #1 expanded
712              ; +---- slot #0 expanded
713              ;
714 02E2  11 FFFF          LD      DE,0FFFFH      ;Initialize lowest address ever found
715 02E5  AF            XOR      A          ;Start from slot #0
716 02E6  4F            LD      C,A          ;Clear bit pattern
717 02E7
718 02E7  D3 A8          CKRM05: OUT     (PPI.AW),A      ;Select the slot
719 02E9  CB 21          SLA     C          ;Shift bit pattern
720 02EB  06 00          LD      B,0          ;Assume this slot is not expanded
721 02ED  21 FFFF          LD      HL,0FFFFH     ;Read from possible expansion slot register
722 02F0  36 F0          LD      (HL),0F0H     ;Write a binary 11110000
723 02F2  7E            LD      A,(HL)
724 02F3  D6 0F          SUB     0FH          ;Read back as 00001111?
725 02F5  20 0B          JR     NZ,CKRM15     ;Nop, this is not an expanded slot
726 02F7  77            LD      (HL),A       ;Write 00000000
727 02F8  7E            LD      A,(HL)
728 02F9  3C            INC     A          ;Read back as 11111111?
729 02FA  20 06          JR     NZ,CKRM15     ;Nop, not expanded slot
730 02FC  04            INC     B          ;We're checking expanded slot
731 02FD  CB C1          SET     0,C          ;Say this slot is expanded
732 02FF
733              CKRM10:
734              ;
735              ; Start from expansion slot #0
736              ;
736 02FF  32 FFFF          LD      (0FFFFH),A    ;Select the expanded slot
737 0302
738 0302  21 BF00          CKRM15: LD      HL,0BF00H ;Start checking from 0BF00H to 8000H
739 0305
740 0305  7E            CKRM20: LD      A,(HL)
741 0306  2F            CPL
```

```

742 0307 77 LD (HL),A
743 0308 BE CP (HL)
744 0309 2F CPL
745 030A 77 LD (HL),A
746 030B 20 07 JR NZ,CKRM25 ;RAM not equipped in this page
747 030D 2C INC L ;Make sure it's not a coincidence
748 030E 20 F5 JR NZ,CKRM20 ;Check more
749 0310 25 DEC H
750 0311 FA 0305 JP M,CKRM20 ;Check next page
751 0314 CKRM25:
752 0314 2E 00 LD L,0
753 0316 24 INC H
754 0317 7D LD A,L ;Below the one ever found
755 0318 93 SUB E
756 0319 7C LD A,H
757 031A 9A SBC A,D
758 031B 30 0A JR NC,CKRM30 ;No
759 031D EB EX DE,HL ;Register this address as the lowest
760 031E 3A FFFF LD A,(0FFFFH) ;Set possible secondary slot #
761 0321 2F CPL
762 0322 6F LD L,A
763 0323 DB A8 IN A,(PPI.AR) ;Set primary slot #
764 0325 67 LD H,A
765 0326 F9 LD SP,HL ;Register these slot #'s
766 0327 CKRM30:
767 0327 78 LD A,B
768 0328 A7 AND A ;Are we checking secondary slot
769 0329 28 0A JR Z,CKRM35 ;No
770 032B 3A FFFF LD A,(0FFFFH)
771 032E 2F CPL
772 032F C6 10 ADD A,10H ;Prepare to select next secondary slot
  
```

```

773 0331 FE 40 CP 01000000B
774 0333 38 CA JR C,CKRM10 ;Continue if more secondary slots remain
775 0335 CKRM35:
776 0335 DB A8 IN A,(PPI.AR)
777 0337 C6 50 ADD A,01010000B ;Prepare to select next slot
778 0339 30 AC JR NC,CKRM05 ;Continue if more primary slots remain
  
```

```

779
780 ;
781 ; Check is done, select the biggest one
782 ;
783 033B 21 0000 LD HL,0
784 033E 39 ADD HL,SP
785 033F 7C LD A,H
786 0340 D3 A8 OUT (PPI.AW),A ;Set primary slot register
787 0342 7D LD A,L
788 0343 32 FFFF LD (0FFFFH),A ;Set possible secondary slot register
789 ;
790 ; Next, check 0C000H..0FFFFH
791 ;
792 0346 79 LD A,C
793 0347 07 RLCA
794 0348 07 RLCA
795 0349 07 RLCA
796 034A 07 RLCA
797 034B 4F LD C,A
798 034C 11 FFFF LD DE,0FFFFH ;Initialize lowest address ever found
799 034F DB A8 IN A,(PPI.AR) ;Start from slot #0
800 0351 E6 3F AND 00111111B
801 0353 CKRM50:
802 0353 D3 A8 OUT (PPI.AW),A ;Select the slot
803 0355 06 00 LD B,0 ;Assume this slot is not expanded
804 0357 CB 01 RLC C ;Shift bit pattern
805 0359 30 0A JR NC,CKRM60 ;This slot is not expanded
806 035B 04 INC B ;We're checking expanded slot
807 035C 3A FFFF LD A,(0FFFFH)
808 035F 2F CPL
809 0360 E6 3F AND 00111111B
  
```

```

( MSX ROM BASIC BIOS ) Macro-80      3.44   01-Jan-85   PAGE   17-1      33
- MSXIO - Find available RAM

810 0362          CKRM55:
811 0362 32 FFFF          LD      (OFFFHH),A      ;Select the expanded slot
812 0365          CKRM60:
813 0365 21 FE00          LD      HL,0FE00H      ;Start checking from 0FE00H to 0C000H
814 0368          CKRM65:
815 0368 7E          LD      A,(HL)
816 0369 2F          CPL
817 036A 77          LD      (HL),A
818 036B BE          CP      (HL)
819 036C 2F          CPL
820 036D 77          LD      (HL),A
821 036E 20 09        JR      NZ,CKRM70      ;RAM not equipped in this page
822 0370 2C          INC     L              ;Make sure it's not a coincidence
823 0371 20 F5        JR      NZ,CKRM65      ;Check more
824 0373 25          DEC     H
825 0374 7C          LD      A,H
826 0375 FE C0        CP      0C0H
827 0377 30 EF        JR      NC,CKRM65      ;Check next page
828 0379          CKRM70:
829 0379 2E 00        LD      L,0
830 037B 24          INC     H
831 037C 7D          LD      A,L              ;Below the one ever found
832 037D 93          SUB     E
833 037E 7C          LD      A,H
834 037F 9A          SBC     A,D
835 0380 30 0A        JR      NC,CKRM75      ;No
836 0382 EB          EX      DE,HL           ;Register this address as the lowest
837 0383 3A FFFF        LD      A,(OFFFHH)      ;Set possible secondary slot #
838 0386 2F          CPL
839 0387 6F          LD      L,A
840 0388 DB A8        IN      A,(PPI.AR)      ;Set primary slot #

```

```

( MSX ROM BASIC BIOS ) Macro-80      3.44   01-Jan-85   PAGE   17-2      34
- MSXIO - Find available RAM

841 038A 67          LD      H,A
842 038B F9          LD      SP,HL           ;Register these slot #'s
843 038C          CKRM75:
844 038C 78          LD      A,B
845 038D A7          AND     A              ;Are we checking secondary slot
846 038E 28 08        JR      Z,CKRM80        ;No
847 0390 3A FFFF        LD      A,(OFFFHH)
848 0393 2F          CPL
849 0394 C6 40        ADD     A,01000000B     ;Prepare to select next secondary slot
850 0396 30 CA        JR      NC,CKRM55      ;Continue if more secondary slots remain
851 0398          CKRM80:
852 0398 DB A8        IN      A,(PPI.AR)
853 039A C6 40        ADD     A,01000000B     ;Prepare to select next slot
854 039C 30 B5        JR      NC,CKRM50      ;Continue if more primary slots remain
855          SUBTTL - MSXIO - Slot attribute setup

```

```

( MSX ROM BASIC BIOS ) Macro-80      3.44   01-Jan-85   PAGE   18      35
- MSXIO - Slot attribute setup

856
857
858          ;
859          ; Check is done, select the biggest one
860          ;
860 039E 21 0000        LD      HL,0
861 03A1 39          ADD     HL,SP
862 03A2 7C          LD      A,H
863 03A3 D3 A8        OUT     (PPI.AW),A      ;Set primary slot register
864 03A5 7D          LD      A,L
865 03A6 32 FFFF        LD      (OFFFHH),A      ;Set possible secondary slot register
866 03A9 79          LD      A,C              ;Set 'slot expanded' flag
867
868          ;
869          ; Clear work area with zero
870          ;
870 03AA 01 0C49        LD      BC,0C49H        ;length of work area
871 03AD 11 F381        LD      DE,RAML0W+1
872 03B0 21 F380        LD      HL,RAML0W      ;beginning of work
873 03B3 36 00        LD      (HL),0          ;init first byte
874 03B5 ED B0        LDIR
875
876          ;
877          ; Set EXPTEL
878          ;
878 03B7 4F          LD      C,A              ;Set 'slot-expanded' flag
879 03B8 06 04        LD      B,4              ;Loop 4 times
880 03BA 21 FCC4        LD      HL,EXPTBL+3
881 03BD          SSLTLP:
882 03BD CB 19          RR      C              ;Set carry if LSB is set
883 03BF 9F          SBC     A,A              ;[Acc]=255 if expanded, 0 if not expanded
884 03C0 E6 80        AND     80H              ;Affects only MSB
885 03C2 77          LD      (HL),A          ;Set table for each slot
886 03C3 2B          DEC     HL

```



```

-- MSXIO - Slot attribute setup

887 03C4 10 F7          DJNZ  SSLTLP
888
889                    ; Set SLTTBL
890
891 03C6 DB A8          IN    A,(PPI.AR) ;Remember primary slot register's content
892 03C8 4F             LD    C,A
893 03C9 AF            XOR    A ;Read from slot #0
894 03CA D3 A8          OUT   (PPI.AW),A
895 03CC 3A FFFF        LD    A,(0FFFFH)
896 03CF 2F            CPL
897 03D0 6F            LD    L,A
898 03D1 3E 40          LD    A,0100000B ;Read from slot #1
899 03D3 D3 A8          OUT   (PPI.AW),A
900 03D5 3A FFFF        LD    A,(0FFFFH)
901 03D8 2F            CPL
902 03D9 67            LD    H,A
903 03DA 3E 80          LD    A,80H ;Read from slot #2
904 03DC D3 A8          OUT   (PPI.AW),A
905 03DE 3A FFFF        LD    A,(0FFFFH)
906 03E1 2F            CPL
907 03E2 5F            LD    E,A
908 03E3 3E C0          LD    A,0C0H ;Read from slot #3
909 03E5 D3 A8          OUT   (PPI.AW),A
910 03E7 3A FFFF        LD    A,(0FFFFH)
911 03EA 2F            CPL
912 03EB 57            LD    D,A
913 03EC 79            LD    A,C ;Restore primary slot register
914 03ED D3 A8          OUT   (PPI.AW),A
915 03EF 22 FCC5        LD    (SLTTBL),HL ;Set SLTTBL
916 03F2 EB            EX    DE,HL
917 03F3 22 FCC7        LD    (SLTTBL+2),HL

```

```

-- MSXIO - Slot attribute setup

918 03F6 ED 56          IM    1 ;IM 1
919 03F8 C3 2680        JP    INIT
920                    SUBTTL - MSXIO - Control-[C] processing

```

```

-- MSXIO - Control-[C] processing

921
922 03FB                ISCNTC:
923 03FB 3A FBBL        LD    A,(BASROM) ;Is BASIC text in ROM
924 03FE A7             AND    A
925 03FF C0             RET    NZ ;Yes
926 0400 E5            PUSH   HL
927 0401 21 FC9B        LD    HL,INTFLG ;Seen any interesting key
928 0404 F3            DI
929 0405 7E            LD    A,(HL)
930 0406 36 00          LD    (HL),0
931 0408 E1            POP    HL
932 0409 FB            EI
933 040A A7             AND    A
934 040B C8             RET    Z ;No
935 040C FE 03          CP    3 ;Is it ctrl-stop?
936 040E 28 1C          JR    Z,EXCABO ;Yes, execution aborted
937
938                    ; Pause until next STOP is pressed
939
940 0410 E5            PUSH   HL ;STOP pressed (pause)
941 0411 D5            PUSH   DE
942 0412 C5            PUSH   BC
943 0413 CD 09DA        CALL  CKDPC0 ;Display cursor if disabled
944 0416 21 FC9B        LD    HL,INTFLG ;Wait for next interesting key
945 0419                WATINT:
946 0419 F3            DI
947 041A 7E            LD    A,(HL)
948 041B 36 00          LD    (HL),0
949 041D FB            EI ;Wait for character if SELECT pressed
950 041E A7             AND    A ;Seen?
951 041F 28 F8          JR    Z,WATINT ;Not yet

```